

Sesión 11 (B) — Regresión simple con datos reales: ajuste y bondad de ajuste en hprice2

Marcos Bujosa

Descripción de la práctica

Retomamos `hprice2` (ya usado en las prácticas de correlación) y sus tres variables etiquetadas: `price`, `rooms`, `nox`. Hoy estimamos, por primera vez con datos *reales*, dos regresiones lineales simples: precio en función del número de habitaciones, y precio en función de la contaminación. Las comparamos entre sí, verificamos las dos condiciones de ortogonalidad y calculamos la bondad de ajuste, conectando con las correlaciones ya calculadas en la práctica de correlación (lección 5) y con el ángulo θ de la lección 8.

Objetivo

1. Estimar dos regresiones lineales simples con datos reales, interpretando el signo y la magnitud de la pendiente sin lenguaje causal.
2. Verificar, con datos reales, las dos condiciones de ortogonalidad de la lección 7.
3. Calcular R^2 “a mano” (STC/SEC/SRC) y compararlo con el que reporta Gretl y con ρ_{xy}^2 .
4. Comparar la bondad de ajuste de los dos modelos y relacionarla con el ángulo θ de la lección 8.
5. Visualizar ambos ajustes mediante diagramas de dispersión con la recta ajustada.

Actividad 1 - Reapertura de hprice2 y etiquetas

en línea de comandos:

```
open hprice2.gdt --quiet
setinfo price -d "Precio mediano de la vivienda ($)" -n "Precio ($)"
setinfo rooms -d "Número medio de habitaciones" -n "Habitaciones"
setinfo nox -d "Concentración de óxidos de nitrógeno" -n "Contaminación (NOx)"
```

Actividad 2 - Regresión: price sobre rooms

en línea de comandos:

```
ols price const rooms
```

- GUI equivalente: Modelo -> Mínimos cuadrados ordinarios, con `price` como dependiente y `const`, `rooms` como regresores.

Licencia: Creative Commons Attribution-ShareAlike 4.0 International (CC BY-SA 4.0).

Interpretación (con cuidado): el coeficiente de `rooms` se lee como “por cada habitación adicional, el precio mediano cambia, en promedio, en esa cantidad de dólares, *manteniendo el resto igual* (aquí, sin ningún otro regresor)”. Evite decir que “una habitación adicional *causa*” ese cambio: como se discutió en la lección 8, la descomposición $STC = SEC + SRC$ es puramente algebraica y no certifica, por sí sola, ninguna relación causal ni su dirección.

Actividad 3 - Regresión: price sobre nox

en línea de comandos:

```
ols price const nox
```

Antes de mirar el resultado: ¿qué signo espera para el coeficiente de `nox`? Contraste su expectativa con lo obtenido.

Actividad 4 - Verificación de las dos condiciones de ortogonalidad

Repetimos ambas regresiones, guardando en cada caso los residuos, para verificar $\sum \hat{e}_i = 0$ y $\sum x_i \hat{e}_i = 0$.

en línea de comandos:

```
ols price const rooms --quiet
series ehat_rooms = $uhat
scalar sum_e_rooms = sum(ehat_rooms)
scalar sum_xe_rooms = sum(rooms*ehat_rooms)

ols price const nox --quiet
series ehat_nox = $uhat
scalar sum_e_nox = sum(ehat_nox)
scalar sum_xe_nox = sum(nox*ehat_nox)

printf "Modelo price-rooms: suma(ehat) = %.8f    suma(rooms*ehat) = %.8f\n", sum_e_rooms, sum_xe_rooms
printf "Modelo price-nox : suma(ehat) = %.8f    suma(nox*ehat) = %.8f\n", sum_e_nox, sum_xe_nox
```

```
Modelo price-rooms: suma(ehat) = -0,00000001    suma(rooms*ehat) = -0,00000019
Modelo price-nox : suma(ehat) = -0,00000001    suma(nox*ehat) = 0,00000005
```

(Los valores no serán exactamente cero por errores de redondeo en coma flotante, pero deben ser insignificantes.)

Actividad 5 - Bondad de ajuste: STC, SEC, SRC y R^2 en ambos modelos

en línea de comandos:

```
ols price const rooms --quiet
scalar mu_price = mean(price)
scalar STC_r = sum((price-mu_price)^2)
scalar SRC_r = sum($uhat^2)
scalar SEC_r = STC_r - SRC_r
scalar R2_r_manual = SEC_r/STC_r

ols price const nox --quiet
scalar STC_n = sum((price-mu_price)^2)
scalar SRC_n = sum($uhat^2)
scalar SEC_n = STC_n - SRC_n
scalar R2_n_manual = SEC_n/STC_n

printf "price-rooms: STC=%.2f SEC=%.2f SRC=%.2f R2(a mano)=%.4f\n", STC_r, SEC_r, SRC_r, R2_r_manual
printf "price-nox : STC=%.2f SEC=%.2f SRC=%.2f R2(a mano)=%.4f\n", STC_n, SEC_n, SRC_n, R2_n_manual
```

```
printf "\nComparacion con rho^2 (recuerde: en regresion simple R2 = rho_xy^2):\n"
printf "corr(price,rooms)^2 = %.4f\n", corr(price,rooms)^2
printf "corr(price,nox)^2 = %.4f\n", corr(price,nox)^2
```

```
price~rooms: STC=42825531146,45 SEC=20732259424,69 SRC=22093271721,76 R2(a mano)=0,4841
price~nox : STC=42825531146,45 SEC=7773157316,47 SRC=35052373829,98 R2(a mano)=0,1815
```

```
Comparacion con rho^2 (recuerde: en regresion simple R2 = rho_xy^2):
corr(price,rooms)^2 = 0,4841
corr(price,nox)^2 = 0,1815
```

Actividad 6 - Diagramas de dispersión con la recta ajustada

- GUI: seleccione price y rooms (con Ctrl), botón derecho, **Gráfico X-Y**; repita con price y nox.
o bien teclee en línea de comandos:

```
gnuplot price rooms
gnuplot price nox
```




Figura 1: Diagramas de dispersión con la recta ajustada.

Compare cómo de “ceñidos” están los puntos que rodean cada recta con el R^2 obtenido en la Actividad 5. Recuerde (lección 8) que $R^2 = \cos^2 \theta$, con θ el ángulo entre los vectores en desviaciones de los datos y del ajuste: cuanto más ceñida la nube a la recta, más pequeño ese ángulo.

Actividad 7 - Síntesis comparativa

Modelo	$\hat{\beta}_0$	$\hat{\beta}_1$	R^2
price~rooms			
price~nox			

Complete la tabla con los valores obtenidos. ¿Cuál de los dos modelos ajusta mejor, según R^2 ? ¿Es coherente con lo observado en los diagramas de dispersión y con las correlaciones ya calculadas en la práctica de correlación (lección 5)?

Preguntas de interpretación para la clase

1. ¿Por qué no podemos afirmar, a partir del modelo `price~rooms`, que “añadir una habitación causa” un aumento de precio?
2. Compare los dos modelos: ¿cuál tiene mayor R^2 ? ¿Qué relación tiene esto con los valores de correlación ya calculados en la práctica de correlación?
3. En los diagramas de dispersión con la recta ajustada, ¿qué modelo muestra una nube de puntos más “pegada” a la recta? ¿Cómo se traduce esto en términos del ángulo θ de la lección 8?
4. ¿Por qué la suma de los residuos no es exactamente cero en la salida de Gretl, sino un número muy pequeño (del orden de 10^{-8} o menor)?
5. Si estimáramos `price` en función de `rooms` y `nox` *simultáneamente* (dos regresores a la vez), ¿por qué NO cabría esperar que el R^2 de ese modelo fuera simplemente la suma de los R^2 de los dos modelos simples de hoy?

Para profundizar:

- Véanse las lecciones 7 y 8 ([S09-Lecc07](#), [S10-Lecc08](#)) para las demostraciones completas de lo que hoy verificamos con datos reales.
- Wooldridge, J. M. (2020). *Introductory Econometrics*, cap. 2, sección 2.4 (advertencias sobre la interpretación causal de R^2).

Código completo de la práctica

```
# -----
# Copyright (C) 2026 Marcos Bujosa
# Licencia: GNU General Public License v3.0 o posterior
# Este código es software libre y puede ser redistribuido y/o modificado bajo los términos de la GPL.
# Ver el archivo LICENSE del repositorio para más detalles.
# -----

# Los dos primeros comandos son necesarios para que Gretl guarde los resultados de la práctica en el
↳ directorio de trabajo
# al ejecutar lo siguiente desde un terminal (use los nombres y ruta que correspondan)
#
#   DIRECTORIO="Nombre_Directorio_trabajo" gretlcli -b ruta/nombre_fichero_de_la_practica.inp
#
# Si esto no le funciona en su sistema, comente las siguientes dos líneas y sitúese en el directorio de
↳ trabajo de gretl
# que corresponda (configure dicho directorio de trabajo desde la ventana principal de Gretl).

string directory = getenv("DIRECTORIO")
set workdir "@directory"

open hprice2.gdt --quiet
setinfo price -d "Precio mediano de la vivienda ($)" -n "Precio ($)"
setinfo rooms -d "Número medio de habitaciones" -n "Habitaciones"
setinfo nox -d "Concentración de óxidos de nitrógeno" -n "Contaminación (NOx)"

ols price const rooms

ols price const nox

outfile --quiet ortogonalidadAmbosModelos.txt
  ols price const rooms --quiet
  series ehat_rooms = $uhat
  scalar sum_e_rooms = sum(ehat_rooms)
  scalar sum_xe_rooms = sum(rooms*ehat_rooms)

  ols price const nox --quiet
  series ehat_nox = $uhat
  scalar sum_e_nox = sum(ehat_nox)
  scalar sum_xe_nox = sum(nox*ehat_nox)

  printf "Modelo price-rooms: suma(ehat) = %.8f   suma(rooms*ehat) = %.8f\n", sum_e_rooms, sum_xe_rooms
  printf "Modelo price-nox : suma(ehat) = %.8f   suma(nox*ehat) = %.8f\n", sum_e_nox, sum_xe_nox
end outfile

outfile --quiet bondadDeAjusteAmbosModelos.txt
  ols price const rooms --quiet
  scalar mu_price = mean(price)
  scalar STC_r = sum((price-mu_price)^2)
  scalar SRC_r = sum($uhat^2)
  scalar SEC_r = STC_r - SRC_r
  scalar R2_r_manual = SEC_r/STC_r

  ols price const nox --quiet
  scalar STC_n = sum((price-mu_price)^2)
  scalar SRC_n = sum($uhat^2)
  scalar SEC_n = STC_n - SRC_n
  scalar R2_n_manual = SEC_n/STC_n

  printf "price~rooms: STC=%.2f SEC=%.2f SRC=%.2f R2(a mano)=%.4f\n", STC_r, SEC_r, SRC_r,
  ↳ R2_r_manual
  printf "price~nox : STC=%.2f SEC=%.2f SRC=%.2f R2(a mano)=%.4f\n", STC_n, SEC_n, SRC_n,
  ↳ R2_n_manual

  printf "\nComparacion con rho^2 (recuerde: en regresion simple R2 = rho_xy^2):\n"
```

```
    printf "corr(price,rooms)^2 = %.4f\n", corr(price,rooms)^2
    printf "corr(price,nox)^2   = %.4f\n", corr(price,nox)^2
end outfile

gnuplot price rooms --output="price_rooms_fit.png"
gnuplot price nox   --output="price_nox_fit.png"
```

Enlace al guión: [S11-Prct-B-hprice2-ajuste.inp](#)

Respuestas

1. ¿Por qué no podemos afirmar causalidad a partir de price~rooms? Porque la descomposición $STC = SEC + SRC$ y el propio ajuste MCO son consecuencias *puramente geométricas* de la ortogonalidad del residuo: se cumplen exactamente igual para cualquier par de vectores $\mathbf{x}, \mathbf{y}$, tengan o no una relación causal real entre sí (recuerde, de la lección 8, el ejemplo del queso y las muertes por enredarse en sábanas). Afirmar causalidad exige una teoría económica previa que justifique la dirección de la relación, algo que la sola regresión no proporciona.
2. Comparación de R^2 entre los dos modelos. El modelo con mayor $|\rho_{\mathbf{x}\mathbf{y}}|$ (en valor absoluto) entre la variable y el regresor tendrá, necesariamente, mayor R^2 , porque en regresión simple $R^2 = \rho_{\mathbf{x}\mathbf{y}}^2$ exactamente (lección 8). Por tanto, el orden de los R^2 de hoy debe coincidir exactamente con el orden, en valor absoluto, de las correlaciones $\rho_{\text{price,rooms}}$ y $\rho_{\text{price,nox}}$ ya calculadas en la práctica de correlación.
3. Nube más pegada a la recta y ángulo θ . El modelo con mayor R^2 es el que tiene la nube de puntos más ceñida a la recta ajustada, porque $R^2 = \cos^2 \theta$: cuanto menor el ángulo θ entre los vectores en desviaciones de los datos y del ajuste, más cerca de 1 está $\cos^2 \theta$, y más “alineados” (en el sentido geométrico de la lección 8) están esos dos vectores —lo que visualmente se traduce en una nube más estrecha alrededor de la recta.
4. Suma de residuos no exactamente cero. Por errores de redondeo en aritmética de coma flotante: los ordenadores no representan los números reales con precisión infinita, así que operaciones que algebraicamente deberían dar cero (como $\sum \hat{e}_i$) arrastran un pequeñísimo error de cálculo, típicamente del orden de 10^{-8} o menor. Esto no contradice la identidad teórica $\sum \hat{e}_i = 0$: es exactamente la misma situación que ya se comprobó en la práctica de verificación geométrica (con datos simulados).
5. R^2 de un modelo con dos regresores no es la suma de los R^2 simples. Porque, como se explicó al cierre de la lección 8, cuando hay más de un regresor no constante el subespacio $\mathcal{L}(\mathbf{1}, \mathbf{x}_1, \mathbf{x}_2)$ centrado tiene dimensión 2, no 1: los vectores en desviaciones de **rooms** y **nox** ya no están alineados en la misma recta, así que el ajuste conjunto $\hat{\mathbf{y}} - \bar{\mathbf{y}}$ ya no es, en general, la suma de las contribuciones “aisladas” de cada regresor. Además, **rooms** y **nox** pueden estar correlacionadas entre sí, de modo que parte de lo que “explica” cada una por separado se solapa. Este fenómeno se estudiará con detalle en la regresión múltiple, más adelante en el curso.